
Abstract—A NeuroXyce synapse device has been implemented.
With conductance-based synaptic dynamics, plasticity deter-
mined by a spike-timing dependent model developed by Clopath-
Gerstner, and a reliability parameter that modulates the trans-
mission of synaptic current as well as the learning process, the
device captures a number of important features that will be useful
in state-of-the-art research on the dynamics of large populations
of spiking neurons. The details of the synapse device will be
discussed.

Index Terms—stdp, clopath-gerstner, phenomenological
synapse model, stochastic transmission reliability

I. INTRODUCTION

For state-of-the-art research into spiking neuron popula-
tion dynamics and learning, it is critical to have a synapse
device that captures some of the primary experimentally
observed features. Thus, we implemented a synapse device
with conductance-based dynamics, spike-timing dependent
plasticity (with long-term potentiation (LTP) and long-term
depression (LTD)), and a stochastic synaptic transmission
reliability modulator. The details of each are discussed in the
sections that follow.

II. BASIC SYNAPSE MODEL WITHOUT PLASTICITY

The model used is based on NEURON simulators Exp2Syn
mechanism [1], [2]. With w representing the Clopath-Gerstner
plasticity scheme outlined in the next section (set to 1 in
the case where no learning occurs), B.V the momentary
postsynaptic voltage, and Erev the reversal potential (set to
−85× 10−3 [V ]), the postsynaptic current is the following:

Ipost = wgMAX(B.V − Erev)

where gMAX, the maximal conductance, is defined as follows:

gMAX = fnorm

(
exp

(
−t
τdecay

)
− exp

(
−t
τrise

))
where fnorm is a normalizing factor that ensures the peak

is 1, τrise is the rise time set at 2× 10−4 [s], and τdecay is the
decay time set to 1×10−2 [s] (making sure that τdecay > τrise).
To run a simulation testing the basic synapse model without
plasticity, please see Netlist VII-A.

III. SPIKE-TIMING DEPENDENT PLASTICITY

We have adapted the Clopath-Gerstner model [3], [4] to be
used in a real-time fashion within a NeuroXyce circuit device
that interacts with Hodgkin-Huxley spiking neuron devices
(the particular model is the standard Hodgkin-Huxley mem-
brane patch model [5]). This well-known phenomenological
model captures a number of the important experimentally
observed behaviors of plasticity in synapses. Additionally, it is
easily tunable to exhibit a variety of STDP curves. The curve
we want to generate is shown in Figure 1. With the following
variables

S = voltage threshold for a spike event
R = voltage value for a resting event
w = weight/strength of synapse

A.V = momentary presynaptic membrane voltage
VL3 = a LPF version of A.V with rate τ3
B.V = momentary postsynaptic membrane voltage
VL1 = a LPF version of B.V with rate τ1
VL2 = a LPF version of B.V with rate τ2

and the Boolean operator on variables x1 and x2 defined as
follows:

(x1 > x2) =

{
1 if x1 > x2

0 else

the modified Clopath/Gerstner equation that updates the
synaptic weight is as follows:

dw

dt
=

(
dwLTD

dt
+
dwLTP

dt

)
(w > wmin)(w < wmax)

where the changes in w due to LTD and LTP are:

dwLTD

dt
= −ALTD(A.V > S)(VL1 > R)(VL1 −R)

dwLTP

dt
= ALTPVL3(B.V > S)(B.V − S)(VL2 > R)(VL2 −R)

while the changes in the LPF voltages are:



dVL1

dt
=

B.V − VL1

τ1
dVL2

dt
=

B.V − VL2

τ2
dVL3

dt
=

(A.V > S)− VL3

τ3

The parameters are set as follows (note that some parameter
values are different from the Clopath-Gerstner papers; this was
necessary to obtain the desired behavior):

S = −45.3× 10−3 [V ]

R = −72.655× 10−3 [V ]

wmin = 0.0

wmax = 1.6

w = 1(initial value)
ALTD = 5× 10−2

[
V −1

]
ALTP = 8.5

[
V −2

]
τ1 = 23× 10−3 [s]

τ2 = 7× 10−3 [s]

τ3 = 46× 10−3 [s]

For an example of the pair of spiking neurons, as well as
the different LPF voltages, and the synaptic dynamics over
time, please see Figure 2. Additionally, to run the simulation,
please see Netlist VII-B.

IV. STOCHASTIC TRANSMISSION RELIABILITY

In experimental studies, action potentials generated in presy-
naptic neuron, only released neurotransmitter to postsynaptic
neurons about 10% of the time [6]. This value can vary
depending on the species of neurons, synapses, etc. However,
it is evident that being able to adjust the synaptic transmission
reliability is imperative. In fact, we conjecture that stochastic
transmission failure at the single synapse level plays a critical
role in enabling the generation of population-level attractor
dynamics that could serve as the basis for a multi-modal
associative memory [7], [8]. Therefore, having such function-
ality in our NeuroXyce device is important. Specifically, this
behavior functions as follows (when a given presynaptic neural
spike event occurs):

• with probability P , the synapse will work as usual
• with probability (1− P ), w will not be updated and no

synaptic current will be generated
In order to confirm functionality, two experiments were

conducted to test that there were the expected number of: (1)
synaptic weight updates N (∆w) and (2) nonzero postsynaptic
currents, N (Ipost 6= 0). In both experiments, P was fixed at a
given value and the simulation was run 100 times; this was
carried out for P ∈ {0, 0.1, 0.2, . . . , 1}. For the purposes of the
experiments, it sufficed to fix the timing between presynaptic
and postsynaptic neurons. In particular, it was set to 10 [ms]

difference, pre-before-post. The results of the experiments are
summarized below.

A. w updates

For each P value, the actual average E [N (∆w)] was
computed and compared to the theoretically expected number
Ê [N (∆w)]. The results are as follows:

P Ê [N (∆w)] E [N (∆w)]
0% 0 0
10% 2 1.8416
20% 4 4.0594
30% 6 6.2178
40% 8 8.1584
50% 10 10.1188
60% 12 11.9901
70% 14 14.0594
80% 16 16.2574
90% 18 18.2475
100% 20 20

The actual agree well with the expected, confirming the
proper function.

B. nonzero Ipost

For each P value, the actual average E [N (Ipost 6= 0)] was
computed and compared to the theoretically expected number
Ê [N (Ipost 6= 0)]. The results are as follows:

P Ê [N (Ipost 6= 0)] E [N (Ipost 6= 0)]
0% 0 0
10% 2 1.8416
20% 4 4.0594
30% 6 6.2178
40% 8 8.1584
50% 10 10.1188
60% 12 11.9901
70% 14 14.0594
80% 16 16.2574
90% 18 18.2475
100% 20 20

The actual agree well with the theoretical, confirming the
proper function. The astute observer will notice that the actual
values are the same in the second experiment as in the first
experiment; clearly, this is what one would expect. For an
example of the pair of spiking neurons, as well as the different
LPF voltages, and the synaptic dynamics over time, when
P = 0.50 for a single simulation, please see Figure 3.
Additionally, to run the simulation, please see Netlist VII-C.

V. CONCLUSION

The device appears to work as desired. It will be used in
the generation of a model intended to investigate the extent to
which the dynamics of a population of spiking neurons can
be used as the basis for an associative memory. Of particular
interest is the role that synaptic transmission failure plays in
the generation of stable states far-from-equilibrium, as they ap-
pear to evolve in actual neocortical regions. We conjecture that



such transmission failure is critical to encoding multi-sensory
memories that are robust to noise and exhibit a high degree
of associativity. In fact, we posit that dynamical distributed
information encoding within a population of spiking neurons
depends on modulating the synaptic transmission reliability. A
number of large-scale real-time experiments will be designed
to investigate these theories.

VI. FIGURES

Fig. 1. With presynaptic neuron fixed in time, the postsynaptic neuron varied its relative timing from -80[ms] (post-before-pre) to 80[ms] (pre-before-post).
16 spike pairs at a given timing would be stimulated; the synaptic weight difference ∆w = w16 −w0 would be computed (where wn represents the weight
after the nth spike). The horizontal axis shows ∆t; the vertical axis shows ∆w.

Fig. 2. 16 spike pairs were stimulated at a given timing; this plot shows when it is post-before-pre 80[ms] (which accounts for the observed long-term
depression of the synaptic weight,w) while the synaptic transmission reliability, P , is 100% (which accounts for the spiking of the postsynaptic neuron and
the updating of the synaptic weight after each presynaptic neural spike). The plot shows various variables of interest that exhibit how the Clopath-Gerstner
STDP model works.



0 1000 2000 3000 4000 5000 6000 7000 8000 9000
−100

−50

0

50

P
re

 [
m

V
]

0 1000 2000 3000 4000 5000 6000 7000 8000 9000
−0.02

0

0.02

0.04

0.06

V
L

3
 [

u
n

it
le

s
s
]

0 1000 2000 3000 4000 5000 6000 7000 8000 9000
−100

−50

0

50

P
o

s
t 

[m
V

]

0 1000 2000 3000 4000 5000 6000 7000 8000 9000
−75

−70

−65

−60

V
L

1
 [

m
V

]

0 1000 2000 3000 4000 5000 6000 7000 8000 9000
−80

−70

−60

−50

−40

V
L

2
 [

m
V

]

0 1000 2000 3000 4000 5000 6000 7000 8000 9000
0.8

1

1.2

1.4

1.6

s
y
n

 w

Fig. 3. 20 presynaptic spikes were stimulated. The postsynaptic neuron, if it spiked, would do so after the presynaptic neuron as its only source of current
was in response to the presynaptic neural spike and delivered via the synapse device (which accounts for the observed long-term potentiation of the synaptic
weight, w) while the synaptic transmission reliability, P , is 50% (which accounts for the spiking of the postsynaptic neuron and the updating of the synaptic
weight in response to about 50% of the presynaptic neural spikes).



VII. NEUROXYCE NETLISTS

A. Basic Synapse Model without plasticity

Test Synapse Level 3 device by Alex Duda

*No learning, perfectly reliable.

*13 August 2012

.options timeint method=7 newlte=1 newbpstepping=1 reltol=1e-4

.GLOBAL_PARAM TIMING=410e-3

.param AMP={3.14159*.1825e-12}

.param WIDTH={1e-3}

.param PERIOD = {400e-3}

*FIRST WE NEED TO SEND A CURRENT INPUT to HH1 and see its effect on HH2.

In11 0 a1 PULSE( 0 {AMP} {400e-3} 1.0e-6 1.0e-6 {WIDTH} {PERIOD} )

*SECOND WE NEED TO ADD HH1 TO RECEIVE INPUT.

.param segLength = 1e-4 ; [cm]

.param segDiameter = 1e-4 ; [cm]

.param segSurfaceArea = { 3.14159 * segDiameter * segLength }

* specific membrane capacitance 1uF/cmˆ2
.param memC = { 1.0e-6 * segSurfaceArea } ; [F]

* leak current has membrane resistivity of 40,000 ohm cmˆ2, with reversal potential of -65mV
.param rm = { 4.0e4 / segSurfaceArea } ; [ohm]
.param memG = { 1 / rm } ; [1/ohm]
.param revE = -0.065 ; [V]

* active conductances

* Na specific conductance is 1200 S/mˆ2 = 1.2e-1 S/cmˆ2
.param gnas = { 0.12 * segSurfaceArea } ; [S]
.param ErevNa = 0.05 ; [V]

* K specific conductance is 360 S/mˆ2 = 3.6e-2 S/cmˆ2
.param gks = { 0.036 * segSurfaceArea } ; [S]
.param ErevK = -0.077 ; [V]

* neuron model
.model HH_Params neuron level=1 cMem={memC} gMem={memG} eLeak={revE} gNa={gnas} gK={gks}
eNa={ErevNa} eK={ErevK} vRest={revE}

*CREATE THREE NEURON INSTANCES

yneuron HH1 a1 0 HH_Params
yneuron HH2 a2 0 HH_Params

.ic v(a1)=-72.655e-3

.ic v(a2)=-72.655e-3

*THIRD WE NEED TO ADD A SYNAPSE TO GO BETWEEN HH1 AND HH2.

*(express all params in [A], [V], [s], etc.)

*Tune maximal conductance, gMax, properly!



*Let gRheo be roughly the least amount of conductance

*that allows a single presyn neuronal spike to cause a postsyn neuronal spike.
.param gRheo=1.318e-12

*Tune N_Neu parameter such that it takes the desired number of presynaptic

*spiking neurons to make a postsynaptic neuron spike.
.param N_Neu=1

.model synParams synapse level=3 vThresh={-45.3e-3} delay={1e-4} gMax={gRheo/N_Neu} eRev={0}
tau1={1e-4} tau2={5e-3} ALTD={5e-2} ALTP={8.5} L1TAU=23e-3 L2TAU=7e-3 L3TAU= 46e-3
R=-72.655e-3 S=-45.3e-3 WINIT=1 WMAX=1 WMIN=1

*The P parameter represents the synapse success probability.

*With probability P it will work as usual,

*With probability (1-P) it will fail to generate a synaptic current

*and the w will fail to update.

ysynapse syn12 a1 a2 synParams P={1}

.tran 0 8.4

.print tran i(In11) v(a1) v(a2) n(y%synapse%syn12_w) n(y%synapse%syn12_vl1)
n(y%synapse%syn12_vl2) n(y%synapse%syn12_vl3)
.end

B. Clopath-Gerstner Plasticity Modulator

Test Synapse Level 3 device by Alex Duda

*Learning turned on but perfectly reliable.

*Confirming STDP learning curve.

*13 August 2012

.options timeint method=7 newlte=1 newbpstepping=1 reltol=1e-4

.GLOBAL_PARAM TIMING=320e-3

.STEP TIMING 320e-3 480e-3 1e-3

.param AMP={3.14159*.1825e-12}

.param WIDTH={1e-3}

.param PERIOD = {400e-3}

*FIRST WE NEED TO SEND A CURRENT INPUT TO HH1 AND HH2.

In11 0 a1 PULSE( 0 {AMP} {400e-3} 1.0e-6 1.0e-6 {WIDTH} {PERIOD} )
In22 0 a2 PULSE( 0 {AMP} {TIMING} 1.0e-6 1.0e-6 {WIDTH} {PERIOD} )

*SECOND WE NEED TO ADD HHs TO RECEIVE INPUT.

.param segLength = 1e-4 ; [cm]

.param segDiameter = 1e-4 ; [cm]

.param segSurfaceArea = { 3.14159 * segDiameter * segLength }

* specific membrane capacitance 1uF/cmˆ2
.param memC = { 1.0e-6 * segSurfaceArea } ; [F]

* leak current has membrane resistivity of 40,000 ohm cmˆ2, with reversal potential of -65mV
.param rm = { 4.0e4 / segSurfaceArea } ; [ohm]



.param memG = { 1 / rm } ; [1/ohm]

.param revE = -0.065 ; [V]

* active conductances

* Na specific conductance is 1200 S/mˆ2 = 1.2e-1 S/cmˆ2
.param gnas = { 0.12 * segSurfaceArea } ; [S]
.param ErevNa = 0.05 ; [V]

* K specific conductance is 360 S/mˆ2 = 3.6e-2 S/cmˆ2
.param gks = { 0.036 * segSurfaceArea } ; [S]
.param ErevK = -0.077 ; [V]

* neuron model
.model HH_Params neuron level=1 cMem={memC} gMem={memG} eLeak={revE} gNa={gnas} gK={gks}
eNa={ErevNa} eK={ErevK} vRest={revE}

*CREATE TWO NEURON INSTANCES

yneuron HH1 a1 0 HH_Params
yneuron HH2 a2 0 HH_Params

.ic v(a1)=-72.655e-3

.ic v(a2)=-72.655e-3

*THIRD WE NEED TO ADD A SYNAPSE TO GO BETWEEN HH1 AND HH2.

*(express all params in [A], [V], [s], etc.)

*Tune maximal conductance, gMax, properly!

*Let gRheo be roughly the least amount of conductance

*that allows a single presyn neuronal spike to cause a postsyn neuronal spike.
.param gRheo=1.318e-12

*Tune N_Neu parameter such that it takes the desired number of presynaptic

*spiking neurons to make a postsynaptic neuron spike.
.param N_Neu=20

*In order to have a smooth curve, we need to decrease ALTD, ALTP.
.model synParams synapse level=3 vThresh={-45.3e-3} delay={1e-4} gMax={gRheo/N_Neu} eRev={0}
tau1={1e-4} tau2={5e-3} ALTD={5e-2} ALTP={8.5} L1TAU=23e-3 L2TAU=7e-3 L3TAU= 46e-3
R=-72.655e-3 S=-45.3e-3 WINIT=1 WMAX=1.6 WMIN=0

*The P parameter represents the synapse success probability.

*With probability P it will work as usual.

*With probability (1-P) it will fail to generate a synaptic current and

*the w will fail to update.

ysynapse syn12 a1 a2 synParams P={1}

*.tran 0 31.6
.tran 0 6.5
.print tran i(In11) v(a1) i(In22) v(a2) n(y%synapse%syn12_w) n(y%synapse%syn12_vl1)
n(y\%synapse\%syn12_vl2) n(y\%synapse\%syn12_vl3)
.end



C. Transmission Probability

Test Synapse Level 3 device by Alex Duda

*(adjusted parameters and plasticity with access to internal states/variables)

*Configured so one spiking presynaptic neuron will make one postsynaptic neuron spike

*ALTP and ALTD tuned for somewhat smooth stdp curve

*13 August 2012

.options timeint method=7 newlte=1 newbpstepping=1 reltol=1e-4

.GLOBAL_PARAM TIMING=410e-3

*.STEP TIMING 390e-3 410e-3 1e-3

.GLOBAL_PARAM TEST=0

.STEP TEST 0 1 0.01

.param S=0.50

.param AMP={3.14159*.1825e-12}

.param WIDTH={1e-3}

.param PERIOD = {400e-3}

*FIRST WE NEED TO SEND A CURRENT PAIR of INPUTS to HH1 and HH2.

In11 0 a1 PULSE( 0 {AMP} {400e-3} 1.0e-6 1.0e-6 {WIDTH} {PERIOD} )

*SECOND WE NEED TO ADD HH1 TO RECEIVE INPUT.

.param segLength = 1e-4 ; [cm]

.param segDiameter = 1e-4 ; [cm]

.param segSurfaceArea = { 3.14159 * segDiameter * segLength }

* specific membrane capacitance 1uF/cmˆ2
.param memC = { 1.0e-6 * segSurfaceArea } ; [F]

* leak current has membrane resistivity of 40,000 ohm cmˆ2, with reversal potential of -65mV
.param rm = { 4.0e4 / segSurfaceArea } ; [ohm]
.param memG = { 1 / rm } ; [1/ohm]
.param revE = -0.065 ; [V]

* active conductances

* Na specific conductance is 1200 S/mˆ2 = 1.2e-1 S/cmˆ2
.param gnas = { 0.12 * segSurfaceArea } ; [S]
.param ErevNa = 0.05 ; [V]

* K specific conductance is 360 S/mˆ2 = 3.6e-2 S/cmˆ2
.param gks = { 0.036 * segSurfaceArea } ; [S]
.param ErevK = -0.077 ; [V]

* neuron model
.model HH_Params neuron level=1 cMem={memC} gMem={memG} eLeak={revE} gNa={gnas} gK={gks}
eNa={ErevNa} eK={ErevK} vRest={revE}

*CREATE THREE NEURON INSTANCES

yneuron HH1 a1 0 HH_Params
yneuron HH2 a2 0 HH_Params



.ic v(a1)=-72.655e-3

.ic v(a2)=-72.655e-3

*THIRD WE NEED TO ADD A SYNAPSE TO GO BETWEEN HH1 AND HH2.

*(express all params in [A], [V], [s], etc.)

*Tune maximal conductance, gMax, properly!

*Let gRheo be roughly the least amount of conductance

*that allows a single presyn neuronal spike to cause a postsyn neuronal spike.
.param gRheo=1.318e-12

*Tune N_Neu parameter such that it takes the desired number of presynaptic

*spiking neurons to make a postsynaptic neuron spike.
.param N_Neu=1

.model synParams synapse level=3 vThresh={-45.3e-3} delay={1e-4} gMax={gRheo/N_Neu} eRev={0}
tau1={1e-4} tau2={5e-3} ALTD={5e-2} ALTP={8.5} L1TAU=23e-3 L2TAU=7e-3 L3TAU= 46e-3
R=-72.655e-3 S=-45.3e-3 WINIT=1 WMAX=1.6 WMIN=0

*The P parameter represents the synapse success probability.

*With probability P it will work as usual.

*With probability (1-P) it will fail to generate a synaptic current and

*the w will fail to update.

ysynapse syn12 a1 a2 synParams P={S}

*.tran 0 31.6
.tran 0 8.4
.print tran i(In11) v(a1) v(a2) n(y%synapse%syn12_w) n(y%synapse%syn12_vl1)
n(y%synapse%syn12_vl2) n(y%synapse%syn12_vl3)
.end


